Save Images as KML Files - Contour Example



In this example we create a map of the world using the Mollweide projection, and overplot that map with two 3-D contour plots (one that displays filled contour levels with different colors, and one that just shows the contour boundaries). We then use the `CONTOUR` function's `SAVE` method to create a KML file and load the file into the Google Earth™ mapping service.

This code creates a KML file that creates the image shown above. You can copy the entire block and paste it into the IDL command line to run it.

```
; Create the contour data

; X-coordinates of the Z data (-180 to 180)
longitude = FINDGEN(360) - 180

; y-coordinates of the Z data (-90 to 90)
latitude = FINDGEN(180) - 90

; A 2-D array containing the contour data
cntrdata = SIN(longitude/30) # COS(latitude/30)

; Create a map graphic as a canvas for the contour plot
worldmap = MAP('Mollweide', LIMIT=[-90,-180,90,180])

; Overplot a ten-level contour plot, where each contour level is
; filled with a color. The levels are 50% opaque and the colors
; are taken from the Rainbow color table
```

```
cntr1 = CONTOUR(cntrdata, longitude, latitude, $
  GRID_UNITS=2, N_LEVELS=10, RGB_TABLE=13, /OVERPLOT, $
  /FILL, TRANSPARENCY=50)

; Overplot another contour plot, this time displaying only
; contour boundaries. These boundaries are two points thick
; and are colored with the Rainbow + White color table
cntr2 = CONTOUR(cntrdata, longitude, latitude, $
  GRID_UNITS=2, N_LEVELS=10, RGB_TABLE=39, $
  /OVERPLOT, C_THICK=[2])

; Save the graphic as a KML file
worldmap.SAVE, 'contour_map.kml'
```

At this point you can load `contour_map.kml` into the Google Earth™ mapping service, and the map and overplotted contour plots are projected on a globe.